Oneway and Twoway ANOVA Models Using SAS
For this example we again use data from the Werner birth control study. Data for this study were collected from 188 women, 
SAS commands to read in the raw data and create a permanent SAS dataset are shown below. Notice that the input statement lists the informat for each variable, e.g., 4.0, which is the SAS W.D format, where W is the total width of the field for the variable and D indicates the number of places after the decimal. The decimals in the raw data will override the .D specification that you give, but can be used to assign the correct number of decimals to a value if none are present in the data. We use cutpoints for AGE to create AGEGROUP,
libname b510 "c:\documents and settings\kwelch\desktop\b510";

DATA b510.WERNER;

  INFILE "WERNER2.DAT";

  INPUT ID 4.0 AGE 4.0 HT 4.0 WT 4.0

        PILL 4.0 CHOL 4.0  ALB 4.1

        CALC 4.1 URIC 4.1  PAIR 3.0;

  if ht=999 then ht=.;

  if wt=999 then wt=.;

  if alb=99 then alb=.;

  if calc=99 then calc=.;

  if uric=99 then uric=.;

  if age not=. then do;

     if age <= 25 then agegroup=1;


 if age  > 25 and age <= 40 then agegroup=2;


 if age  > 40 then agegroup=3;

  end;

  if chol >=600 then chol=.;

  if chol <=50  then chol=.;

run;
We create user-defined formats for PILL and AGEGROUP. We will use a Format Statement to assign these formats to the appropriate variables in each Proc Step.
proc format;

  value pillfmt 1="No Pill"

                2="Pill";

  value agefmt  1="Young"

                2="Middle"



    3="Mature";

run;
Next, we get descriptive statistics for CHOL for each level of PILL and for each level of AGEGROUP, and look at Boxplots for CHOL for each level of these variables.
proc means data=b510.werner;

  class pill;

  format pill pillfmt.;

  var chol;

run;
proc means data=b510.werner;

  class agegroup;

  format agegroup agefmt.;

  var chol;

run;
proc sgplot data=b510.werner;

   vbox chol / category=pill;

format pill pillfmt.;

run;
proc sgplot data=b510.werner;

   vbox chol / category=agegroup;

format agegroup agefmt.;

run;

                                      The MEANS Procedure

                                   Analysis Variable : CHOL

                  N

     PILL       Obs      N            Mean         Std Dev         Minimum         Maximum

     -------------------------------------------------------------------------------------

     No Pill     94     94     232.9680851      43.4915520     155.0000000     335.0000000

     Pill        94     92     239.4021739      41.5620970     160.0000000     390.0000000

     -------------------------------------------------------------------------------------
                                  Analysis Variable : CHOL

                  N

    agegroup    Obs      N            Mean         Std Dev         Minimum         Maximum

    --------------------------------------------------------------------------------------

    Young        52     50     222.1200000      37.4441843     155.0000000     330.0000000

    Middle       84     84     229.3928571      38.4397566     160.0000000     317.0000000

    Mature       52     52     260.5576923      44.0656433     160.0000000     390.0000000

    --------------------------------------------------------------------------------------

We now use Proc Sgpanel to look at histograms and boxplots for CHOL for each combination of levels of AGEGROUP and PILL, so we can see how these variables jointly relate to cholesterol levels.

proc sgpanel data=b510.werner;

  panelby agegroup / columns=3 novarname;

  vbox chol / category=pill;

format agegroup agefmt.;

format pill pillfmt.;

run;

proc sgpanel data=b510.werner;

  panelby agegroup pill / columns=2 rows=3;

  histogram chol;

format agegroup agefmt.;

format pill pillfmt.;

run;

We now use Proc GLM to fit a oneway ANOVA model to compare the means of CHOL for each level of PILL and another oneway ANOVA model to compare the means of CHOL for each level of AGEGROUP.

/*Oneway ANOVA Model for Pill*/

proc glm data=b510.werner order=internal;

  class pill;

  format pill pillfmt.;

  model chol = pill;

  means pill / hovtest=levene(type=abs) ;

run; quit;
                                     The GLM Procedure

                                    Class Level Information

                             Class         Levels    Values

                             PILL               2    No Pill Pill

                            Number of Observations Read         188

                            Number of Observations Used         186

                                       The GLM Procedure

Dependent Variable: CHOL

                                              Sum of

      Source                      DF         Squares     Mean Square    F Value    Pr > F

      Model                        1       1924.7611       1924.7611       1.06    0.3038

      Error                      184     333105.0238       1810.3534

      Corrected Total            185     335029.7849

                      R-Square     Coeff Var      Root MSE     CHOL Mean

                      0.005745      18.01743      42.54825      236.1505

      Source                      DF       Type I SS     Mean Square    F Value    Pr > F

      PILL                         1     1924.761126     1924.761126       1.06    0.3038

      Source                      DF     Type III SS     Mean Square    F Value    Pr > F

      PILL                         1     1924.761126     1924.761126       1.06    0.3038

                        Levene's Test for Homogeneity of CHOL Variance

                         ANOVA of Absolute Deviations from Group Means

                                       Sum of        Mean

                 Source        DF     Squares      Square    F Value    Pr > F

                 PILL           1       949.6       949.6       1.49    0.2233

                 Error        184      117039       636.1

                       Level of            -------------CHOL------------

                       PILL          N             Mean          Std Dev

                       No Pill      94       232.968085       43.4915520

                       Pill         92       239.402174       41.5620970

/*Oneway ANOVA for AGEGROUP*/

proc glm data=b510.werner order=internal;

  class agegroup;

  format agegroup agefmt.;

  model chol = agegroup;

  means agegroup;

  means agegroup / hovtest=levene(type=abs) tukey bon scheffe dunnett("Young") ;

run; quit;

                                       The GLM Procedure

                                   Class Level Information

                         Class         Levels    Values
                         agegroup           3    Young Middle Mature

                            Number of Observations Read         188

                            Number of Observations Used         186

Dependent Variable: CHOL

                                              Sum of

      Source                      DF         Squares     Mean Square    F Value    Pr > F
      Model                        2      44655.6423      22327.8212      14.07    <.0001

      Error                      183     290374.1426       1586.7439
      Corrected Total            185     335029.7849

                      R-Square     Coeff Var      Root MSE     CHOL Mean
                      0.133289      16.86803      39.83395      236.1505

      Source                      DF       Type I SS     Mean Square    F Value    Pr > F
      agegroup                     2     44655.64231     22327.82115      14.07    <.0001

      Source                      DF     Type III SS     Mean Square    F Value    Pr > F
      agegroup                     2     44655.64231     22327.82115      14.07    <.0001

                       Level of            -------------CHOL------------

                       agegroup      N             Mean          Std Dev

                       Young        50       222.120000       37.4441843

                       Middle       84       229.392857       38.4397566

                       Mature       52       260.557692       44.0656433

                       Levene's Test for Homogeneity of CHOL Variance

                         ANOVA of Absolute Deviations from Group Means

                                        Sum of        Mean

                Source          DF     Squares      Square    F Value    Pr > F

                agegroup         2       460.2       230.1       0.43    0.6530

                Error          183     98572.1       538.6

                         Tukey's Studentized Range (HSD) Test for CHOL

                NOTE: This test controls the Type I experimentwise error rate.

                         Alpha                                   0.05

                         Error Degrees of Freedom                 183

                         Error Mean Square                   1586.744

                         Critical Value of Studentized Range  3.34170

                Comparisons significant at the 0.05 level are indicated by ***.

                                       Difference

                       agegroup           Between     Simultaneous 95%

                      Comparison            Means    Confidence Limits

                    Mature - Middle        31.165      14.556   47.773  ***

                    Mature - Young         38.438      19.795   57.081  ***

                    Middle - Mature       -31.165     -47.773  -14.556  ***

                    Middle - Young          7.273      -9.540   24.085

                    Young  - Mature       -38.438     -57.081  -19.795  ***

                    Young  - Middle        -7.273     -24.085    9.540

                              Bonferroni (Dunn) t Tests for CHOL

 NOTE: This test controls the Type I experimentwise error rate, but it generally has a higher

                 Type II error rate than Tukey's for all pairwise comparisons.

                               Alpha                        0.05

                               Error Degrees of Freedom      183

                               Error Mean Square        1586.744

                               Critical Value of t       2.41619

                Comparisons significant at the 0.05 level are indicated by ***.

                                       Difference

                       agegroup           Between     Simultaneous 95%

                      Comparison            Means    Confidence Limits

                    Mature - Middle        31.165      14.182   48.148  ***

                    Mature - Young         38.438      19.374   57.501  ***

                    Middle - Mature       -31.165     -48.148  -14.182  ***

                    Middle - Young          7.273      -9.919   24.464

                    Young  - Mature       -38.438     -57.501  -19.374  ***

                    Young  - Middle        -7.273     -24.464    9.919

                                    Scheffe's Test for CHOL

 NOTE: This test controls the Type I experimentwise error rate, but it generally has a higher

                 Type II error rate than Tukey's for all pairwise comparisons.

                               Alpha                        0.05

                               Error Degrees of Freedom      183

                               Error Mean Square        1586.744

                               Critical Value of F       3.04531

                Comparisons significant at the 0.05 level are indicated by ***.

                                       Difference

                       agegroup           Between     Simultaneous 95%

                      Comparison            Means    Confidence Limits

                    Mature - Middle        31.165      13.818   48.511  ***

                    Mature - Young         38.438      18.966   57.909  ***

                    Middle - Mature       -31.165     -48.511  -13.818  ***

                    Middle - Young          7.273     -10.287   24.832

                    Young  - Mature       -38.438     -57.909  -18.966  ***

                    Young  - Middle        -7.273     -24.832   10.287

Finally, we use Proc GLM to fit a twoway factorial ANOVA model, with the main effects of CHOL and AGEGROUP and their interaction in the model. We use ODS GRAPHICS in addition to get some diagnostic plots. Notice that in this model, the type I and type III sums of squares are different.
/*Twoway ANOVA model for PILL and AGEGROUP*/

ods graphics on;

proc glm data=b510.werner order=internal plots=(diagnostics);

  class agegroup pill;

  format pill pillfmt.;

  format agegroup agefmt.;

  model chol = pill agegroup pill*agegroup;

  lsmeans pill*agegroup / adjust=tukey slice=agegroup slice=pill;

run; quit;

ods graphics off;
                                       The GLM Procedure

                                   Class Level Information

                         Class         Levels    Values
                         agegroup           3    Young Middle Mature

                         PILL               2    No Pill Pill

                            Number of Observations Read         188

                            Number of Observations Used         186

Dependent Variable: CHOL

                                              Sum of

      Source                      DF         Squares     Mean Square    F Value    Pr > F
      Model                        5      50934.2643      10186.8529       6.45    <.0001

      Error                      180     284095.5206       1578.3084
      Corrected Total            185     335029.7849

                      R-Square     Coeff Var      Root MSE     CHOL Mean
                      0.152029      16.82314      39.72793      236.1505

      Source                      DF       Type I SS     Mean Square    F Value    Pr > F

      PILL                         1      1924.76113      1924.76113       1.22    0.2709

      agegroup                     2     44479.87891     22239.93946      14.09    <.0001

      agegroup*PILL                2      4529.62430      2264.81215       1.43    0.2408

      Source                      DF     Type III SS     Mean Square    F Value    Pr > F

      PILL                         1       672.19671       672.19671       0.43    0.5148

      agegroup                     2     44612.68110     22306.34055      14.13    <.0001

      agegroup*PILL                2      4529.62430      2264.81215       1.43    0.2408

                                      Least Squares Means

                       Adjustment for Multiple Comparisons: Tukey-Kramer

                                                                 LSMEAN

                        agegroup    PILL        CHOL LSMEAN      Number

                        Young       No Pill      221.653846           1

                        Young       Pill         222.625000           2

                        Middle      No Pill      221.071429           3

                        Middle      Pill         237.714286           4

                        Mature      No Pill      263.500000           5

                        Mature      Pill         257.615385           6

                         Least Squares Means for effect agegroup*PILL

                             Pr > |t| for H0: LSMean(i)=LSMean(j)

                                   Dependent Variable: CHOL

   i/j              1             2             3             4             5             6

      1                      1.0000        1.0000        0.5864        0.0027        0.0163

      2        1.0000                      1.0000        0.6747        0.0048        0.0260

      3        1.0000        1.0000                      0.3935        0.0004        0.0040

      4        0.5864        0.6747        0.3935                      0.1023        0.3422

      5        0.0027        0.0048        0.0004        0.1023                      0.9947

      6        0.0163        0.0260        0.0040        0.3422        0.9947

                                      Least Squares Means

                       agegroup*PILL Effect Sliced by agegroup for CHOL

                                        Sum of

            agegroup        DF         Squares     Mean Square    F Value    Pr > F

            Young            1       11.770385       11.770385       0.01    0.9313

            Middle           1     5816.678571     5816.678571       3.69    0.0565

            Mature           1      450.173077      450.173077       0.29    0.5940

                                      Least Squares Means

                         agegroup*PILL Effect Sliced by PILL for CHOL

                                       Sum of

            PILL           DF         Squares     Mean Square    F Value    Pr > F

            No Pill         2           33510           16755      10.62    <.0001

            Pill            2           15500     7749.884645       4.91    0.0084

